In[1]:=
```
A = {{a, fa}, {b, fb}, {c, fc}}
InterpolatingPolynomial[A, x]
```

Out[1]= {{a, fa}, {b, fb}, {c, fc}}

Out[2]= $$fa + (-a + x)\left(\frac{-fa + fb}{-a + b} + \left(\left(-\frac{-fa + fb}{-a + b} + \frac{-fb + fc}{-b + c}\right)(-b + x)\right) \Big/ (-a + c)\right)$$

In[11]:=
```
L = InterpolatingPolynomial[A, x];
L /. {x → a}
L /. {x → b}
FullSimplify[L /. {x → c}]
```

Out[12]= fa

Out[13]= fb

Out[14]= fc

In[17]:=
```
Li = Integrate[L, {x, a, c}]
```

Out[17]= $$-\Big(\Big((a - c)\left(c^2 (-fa + fb) + a^2 (fb - fc) - 3 b^2 (fa + fc) + 2 b c (2 fa + fc) + 2 a (-c (fa + fb + fc) + b (fa + 2 fc))\right)\Big) \Big/ \big(6 (a - b)(b - c)\big)\Big)$$

In[24]:=
```
FullSimplify[Li /. {b → (a + c) / 2}]
```

Out[24]= $$-\frac{1}{6}(a - c)(fa + 4 fb + fc)$$

In[29]:=
```
Integrate[(x - a) (x - b) (x - c), {x, a, c}]
Integrate[(x - a) (x - b) (x - c), {x, a, c}] /. {b → (a + c) / 2}
```

Out[29]= $$\frac{1}{12}(a - c)^3 (a - 2 b + c)$$

Out[30]= 0

In[28]:=
```
FullSimplify[Integrate[(x - a) (x - b), {x, a, b}]]
```

Out[28]= $$\frac{1}{6}(a - b)^3$$

In[98]:=
```
s = NDSolve[{y'[x] == y[x] Cos[x + y[x]], y[0] == 1}, y, {x, 0, 30}]
```

Out[98]= {{y → InterpolatingFunction[Domain: {{0., 30.}} Output: scalar]}}

In[99]:=
```
Plot[Evaluate[y[x] /. s], {x, 0, 30}, PlotRange → All]
ParametricPlot[Evaluate[{y[x], y'[x]} /. s], {x, 0, 20}]
```

Out[99]=



Out[100]=



In[106]:=
```
s = NDSolve[{y''[x] == -0.2 * y[x], y[0] == 0.3, y'[0]  == 0.0}, y[x], {x, 0, 30}]
```

Out[106]=
```
{{y[x] → InterpolatingFunction[ Domain: {{0., 30.}} Output: scalar ][x]}}
```

In[109]:=
```
Plot[Evaluate[y[x] /. s], {x, 0, 30}, PlotRange → All]
```

Out[109]=



In[110]:=
```
s = NDSolve[{y''[x] == -3.2 * Sin[y[x]] + 0.5 Cos[y[x]] ,
    y[0] == 0.3, y'[0] == 0.0}, y[x], {x, 0, 30}]
Plot[Evaluate[y[x] /. s], {x, 0, 30}, PlotRange → All]
```

Out[110]= {{y[x] → InterpolatingFunction[Domain: {{0., 30.}} Output: scalar][x]}}

Out[111]=



In[84]:=

In[112]:=
```
s = NDSolve[{y''[x] + Sin[y[x]] y[x] == 0, y[0] == 1, y'[0] == 0}, y, {x, 0, 30}]
```

Out[112]= {{y → InterpolatingFunction[Domain: {{0., 30.}} Output: scalar]}}

In[113]:=
```
Plot[Evaluate[{y[x], y'[x], y''[x]} /. s], {x, 0, 30}, PlotStyle -> Automatic]
```

Out[113]=



In[115]:=
```
s = NDSolve[
  {x'[t] == -y[t] - x[t]^2, y'[t] == 2 x[t] - y[t]^3, x[0] == y[0] == 1}, {x, y}, {t, 0, 20}]
```

Out[115]=
```
{{x → InterpolatingFunction[ Domain: {{0., 20.}} Output: scalar ],
  y → InterpolatingFunction[ Domain: {{0., 20.}} Output: scalar ]}}
```

In[116]:=
```
ParametricPlot[Evaluate[{x[t], y[t]} /. s], {t, 0, 20}]
```

Out[116]=



In[117]:=
```
NDSolve[{D[u[t, x], t] == D[u[t, x], x, x], u[0, x] == 0, u[t, 0] == Sin[t], u[t, 5] == 0},
 u, {t, 0, 10}, {x, 0, 5}]
Plot3D[Evaluate[u[t, x] /. %], {t, 0, 10}, {x, 0, 5}, PlotRange → All]
```

Out[117]=
```
{{u → InterpolatingFunction[ Domain: {{0., 10.}, {0., 5.}} Output: scalar ]}}
```

Out[118]=



In[119]:=
```
E1 = -0.13;
E2 = 0.66;
E3 = 1.353;
A12 = 4.2;
A13 = 0.7;
A23 = 0.1;
A21 = A12;
A32 = A23;
A31 = A13;
w = (E2 - E1) * 10;
tmax = 30;
θ = Pi * 0.0;
s = NDSolve[{I x'[t] == E1 x[t] + A12 Sin[w t + θ] y[t] + A13 Sin[w t + θ] z[t],
   I y'[t] == E2 y[t] + A21 Sin[w t + θ] x[t] + A23 Sin[w t + θ] z[t],
   I z'[t] == E3 z[t] + A31 Sin[w t + θ] x[t] + A32 Sin[w t + θ] y[t],
   x[0] == 1, y[0] == 0, z[0] == 0}, {x[t], y[t], z[t]}, {t, 0, 400} ]
Plot[Evaluate[Abs[x[t]]]^2 /. s, {t, 0, tmax}, PlotRange → All]
Plot[Evaluate[Abs[y[t]]]^2 /. s, {t, 0, tmax}, PlotRange → All]
Plot[Evaluate[Abs[z[t]]]^2 /. s, {t, 0, tmax}, PlotRange → All]
Plot[Evaluate[(Abs[x[t]]^2 + Abs[y[t]^2)] /. s, {t, 0, 20}, PlotRange → All]
```

Out[131]= {{x[t] → InterpolatingFunction[Domain: {{0., 400.}} Output: scalar][t],

y[t] → InterpolatingFunction[Domain: {{0., 400.}} Output: scalar][t],

z[t] → InterpolatingFunction[Domain: {{0., 400.}} Output: scalar][t]}}

Out[132]=



Out[133]=



Out[134]=



Out[135]=



In[136]:=

```
Manipulate[Module[{s1, s2},
  s1 = GetLorenzSolution[tfin, β, ρ, σ, {x0, y0, z0}];
  s2 = GetLorenzSolution[tfin, β, ρ, σ, {x0 + xoff, y0 + yoff, z0 + zoff}];
  Grid[
   {{ParametricPlot3D[{{x[t], y[t], z[t]} /. s1, {x[t], y[t], z[t]} /. s2}, {t, 0, tfin},
      ImageSize → {250, 250}, AxesLabel → {x, y, z}]}, {Plot[funcdiff[s1, s2, t],
      {t, 0, tfin}, ImageSize → {230, 230}, AxesLabel → {t, "proximity"}]}}]]
 ,
 "                                                                                ",
 {{tfin, 10, "maximum t"}, 1, 100, .1, Appearance → "Labeled", ImageSize → Tiny},
 {{σ, 10}, 0, 80, 1, Appearance → "Labeled", ImageSize → Tiny},
 {{ρ, 28}, 0, 100, .01, Appearance → "Labeled", ImageSize → Tiny},
 {{β, 8/3}, 0, 4, .1, Appearance → "Labeled", ImageSize → Tiny},
 {{x0, 10, "x(t=0)"}, 0, 100, 1, Appearance → "Labeled", ImageSize → Tiny},
 {{xoff, 0, "x-offset"}, -1, 1, .1, Appearance → "Labeled", ImageSize → Tiny},
 {{y0, 10, "y(t=0)"}, 0, 100, 1, Appearance → "Labeled", ImageSize → Tiny},
 {{yoff, 0, "y-offset"}, -1, 1, .1, Appearance → "Labeled", ImageSize → Tiny},
 {{z0, 10, "z(t=0)"}, 0, 100, .1, Appearance → "Labeled", ImageSize → Tiny},
 {{zoff, 0, "z-offset"}, -1, 1, .0000001, Appearance → "Labeled", ImageSize → Tiny},
 Initialization :→ (GetLorenzSolution[tfin_, β_, ρ_, σ_, rstart_] :=
    Module[{lorenz},
     lorenz = {x'[t] == σ * (y[t] - x[t]),
       y'[t] == x[t] * (ρ - z[t]) - y[t], z'[t] == x[t] * y[t] - β * z[t],
       x[0] == rstart[[1]], y[0] == rstart[[2]], z[0] == rstart[[3]]};
     NDSolve[lorenz, {x[t], y[t], z[t]}, {t, 0, tfin}, MaxSteps → 10^6]];
   funcdiff[s1_, s2_, ta_] := Norm[Module[{}, {x[t], y[t], z[t]} /. s1] -
       Module[{}, {x[t], y[t], z[t]} /. s2]] /. t → ta),
 ControlPlacement → Left, SaveDefinitions → True, TrackedSymbols → All,
 SynchronousUpdating → False]
```

Out[136]=


maximum t 25.4
σ 43
ρ 73.02
β 8/3
x(t=0) 10
x–offset 0
y(t=0) 10
y–offset 0
z(t=0) 10
z–offset 0
y
x
z
proximity
t